

・コンデンサとそれに使われる誘電体材料の温度特性測定プログラムの例

BASIC 言語で記述してあります。

```
1000 ' file "ct4192a.bas",a
1001 ' Copyright (c) 1989-1992 by Osamu Furukawa
1002 ' All Rights Reserved
1004 '
1005 ' for PASOPIA1600 (TS-300E)
1006 '
1010 ON ERROR GOTO 9000
1020 WIDTH 80
1030 KEY OFF
1040 CLEAR ,&HF000
1050 ' WIDTH "LPT1:",255
1060 ' LPRINT CHR$(&H1B);"Q";
1070 '
1080 DEFINT G,I,J,K,L
1090 BLOAD"GPIBDV",&HF000 ← GP-IB を使用可能にするための準備をします。
1100 DIM BF(256)
1110 GPIB=&HF000
1120 GCPRM=&HF100
1130 GBUFF=&HF200
1131 GBINT=0
1132 GBIFC=1
1133 GBRMT=2
1134 GBLLO=3
1135 GBLCL=4
.
(途中省略しています。ここは、GP-IB の初期設定になります。)
.
1180 '
1190 DIM NN$(20),F(5),THICK(20) ← ここは測定のパラメータなどを設定しています。
1200 '
1210 N=1 : NF=5 : TT=0
1220 '
1230 CLS : INPUT"file name (6 letters)";N$ : PRINT : IF LEN(N$)>6 THEN
N$=LEFT$(N$,6)
1240 INPUT"electrode diameter(1)/area(2) ";SE$
1250 IF SE$="2" THEN INPUT" area (cm2) ";AREA : PRINT : GOTO
1290
1260 IF SE$="1" THEN INPUT" diameter(mm) ";DIA : PRINT
1270 AREA=DIA*DIA*.007854
1280 IF SE$<>"1" AND SE$<>"2" THEN GOTO 1240
1290 '
1300 FOR I=1 TO N
1310 PRINT"sample name";
1320 PRINT USING" ##";I;
1330 INPUT NN$(I)
1340 NEXT I : PRINT
1350 FOR I=1 TO N
1360 PRINT"thickness (mm) ";
```

```

1370 PRINT USING"## ";I;
1380 INPUT THICK(I)
1390 NEXT I : PRINT
1400 '
1410 PRINT"select reference temperature." : INPUT"(1)25'C(EIA,MIL)
(2)20'C(JIS,EIAJ) ";SR$: PRINT
1420 IF SR$="1" THEN RT=25 : GOTO 1460
1430 IF SR$="2" THEN RT=20 : GOTO 1460
1440 GOTO 1410
1450 '
1460 FOR I=1 TO NF
1470 F(I)=10^(I-3)
1480 NEXT I
1481 '
1482 INPUT"lower temperature limit ('C) ";TL :
TL=SGN(TL)*2.5*(INT(ABS(TL)/2.5))
1483 INPUT"upper temperature limit ('C) ";TU :
TU=SGN(TU)*2.5*(INT(ABS(TU)/2.5))
1484 NT=(TU-TL)/2.5+1
1485 '
1489 PRINT
1490 INPUT"--- Are you ready(Y/N)";QQ$ ← ここまでが最初の設定です。
1500 IF QQ$="Y" OR QQ$="y" THEN GOTO 1600 ELSE GOTO 1220
1600 '
1610 FOR L=1 TO NF
1620 NZ$=N$+"F"+RIGHT$(STR$(L),1)
1630 OPEN"A:"+NZ$+".DAT" FOR OUTPUT AS #1
1640 PRINT#1,N$
1650 PRINT#1,AREA
1660 PRINT#1,N
1670 PRINT#1,NT
1680 FOR I=1 TO N
1690 PRINT#1,NN$(I)
1700 NEXT I
1710 FOR I=1 TO N
1720 PRINT#1,THICK(I)
1730 NEXT I
1731 PRINT#1,RT
1732 FOR I=1 TO N
1733 PRINT#1,CP1
1734 PRINT#1,DP1
1735 NEXT I
1740 CLOSE 1
1750 NEXT L
1760 '
1770 PRINT
1780 PRINT " Then,let's start!" : PRINT
1790 INPUT " Are you ready(Y/N)";QQ$ ← ここからが、実際の制御と測定になります。
1800 IF QQ$="Y" OR QQ$="y" THEN GOTO 2000 ELSE GOTO 1220
2000 '
2010 ' device 17--- HP4192A
2020 ' 7--- TGP-120-T(MC-81)
2030 PRINT
2040 PRINT" ----- Measurement Start -----"

```

```

2050 FOR I=0 TO 2000 : NEXT I
2060 CLS
2070 GOSUB 8000 ← 自動測定中にディスプレイに表示させる内容です。
2080 '
2090 ' C,D,AUTO,INT,Format 1
2100 GDN=17 : WRT$="A4B2C1T1F1" : GOSUB 5000
2110 ' initial frequency (kHz)
2120 FR=1
2130 FR$="FR"+STR$(FR)+"EN"
2140 ' set frequency
2150 GDN=17 : WRT$=FR$ : GOSUB 5000
2160 ' initial osc level (V)
2170 V=1
2180 OL$="OL"+STR$(V)+"EN"
2190 ' set osc level
2200 GDN=17 : WRT$=OL$ : GOSUB 5000
2210 ' set DC bias off
2220 GDN=17 : WRT$="I0" : GOSUB 5000
2230 GOTO 2290
2240 BI=0
2250 BI$="BI"+STR$(BI)+"EN"
2260 ' set DC bias
2270 GDN=17 : WRT$=BI$ : GOSUB 5000
2280 '
2290 K=1 : TS=TL : ' TS=2.5*(K-1)+TL
2300 ' time signal N0.1-3 ON compressor . remote
2310 GDN=7 : WRT$="ON1.2.3" : GOSUB 5000
2320 TS=INT(TS*10)/10
2330 LOCATE 1,1 : PRINT"      "
2340 LOCATE 3,1 : PRINT"      "
2350 TA$="TA"+STR$(TS)
2360 ' set temperature
2370 GDN=7 : WRT$=TA$ : GOSUB 5000
2380 ' read setting temperature value
2390 GDN=7 : WRT$="!" : GOSUB 5000
2400 GDN=7 : GOSUB 6500 : ' trigger
2410 GDN=7 : GL=6 : GOSUB 6000 : TR$=RED$
2420 TR=VAL(LEFT$(TR$,6))
2430     IF ABS(TS-TR)>.01 THEN PRINT "ERROR":STOP
2440 ' read actual temperature value
2450 GDN=7 : WRT$="MA" : GOSUB 5000
2460 GDN=7 : GOSUB 6500
2470 GDN=7 : GL=6 : GOSUB 6000 : TP$=RED$
2480 TP=VAL(LEFT$(TP$,6))
2490 '
2500 LOCATE 4,21: PRINT USING"####.# ";TS
2510 LOCATE 6,21: PRINT USING"####.# ";TR
2520 LOCATE 8,21: PRINT USING"####.# ";TP
2530 IF ABS(TR-TP)>1! THEN GOTO 2380 ELSE GOTO 2540
2540 ' measurement
2550 FOR I=0 TO 2000 : NEXT I
2551 GOSUB 7500 ← 必要な時間待ちを行い、実際の測定を行っています。
2552 GOSUB 8110 ← 画面表示を行います。
2560 '

```

```

2570 FOR J=1 TO NF
2580   NZ$=N$+"F"+RIGHT$(STR$(J),1)
2590   OPEN "A:"+NZ$+".DAT" FOR APPEND AS #1 ← ファイルに追加書き込みします。
2600   PRINT#1,TS
2610   FOR I=1 TO N
2620     FR$="FR"+STR$(F(J))+".EN"
2630     GDN=17 : WRT$=FR$ : GOSUB 5000
2640     FOR D=0 TO 1500 : NEXT D
2650     ' GDN=17 : GOSUB 6500 : ' trigger ← 測定したデータを受け取ります。
2652     CALL GPIB(GBINT)
2654     CALL GPIB(GBIFC)
2656     CALL GPIB(GBRMT)
2660     FOR D=0 TO 500 : NEXT D
2670     GDN=17 : GL=43 : GOSUB 6000 : CD$=RED$
2680     FOR D=0 TO 500 : NEXT D
2690     ' GDN=17 : GOSUB 6500
2692     CALL GPIB(GBINT)
2694     CALL GPIB(GBIFC)
2696     CALL GPIB(GBRMT)
2700     GDN=17 : GL=43 : GOSUB 6000 : CD$=RED$
2710     DS$=LEFT$(CD$,1) : ' data status
2720     CP=VAL(MID$(CD$,5,11))
2730     DP=VAL(MID$(CD$,21,11))
2740     FP=VAL(MID$(CD$,34,10))
2750     '
2760     CP1=CP*1E+12 : DP1=DP*100
2770     LOCATE 10,21 : PRINT NN$(I)
2780     LOCATE 14,21 : PRINT FP
2790     LOCATE 18,21 : PRINT USING"#####.# ";CP1
2800     LOCATE 20,21 : PRINT USING"   #####.## ";DP1
2810     LOCATE 16,25 : PRINT DS$
2820     '
2830     ' LPRINT USING"#####.# ";TS; ← データ数値を印字します。
2840     ' LPRINT USING"#####.# ";TR;
2850     ' LPRINT USING"#####.# ";TP;
2860     ' LPRINT USING"##.##### ";FP;
2870     ' LPRINT USING"#####.# ";CP1;
2880     ' LPRINT USING"#####.## ";DP1;
2890     ' LPRINT USING"&&";DS$
2900     IF ABS(F(J)-FP)>.1 THEN PRINT"ERROR" : STOP
2910     PRINT#1,CP1
2920     PRINT#1,DP1
2930     FOR D=0 TO 2000 : NEXT D
2940   NEXT I
2950   CLOSE 1
2960 NEXT J
2965 GOSUB 8510
2970 '
2980 K=K+1 : TS=2.5*(K-1)+TL ← 次の測定する温度を設定します。
2985   IF (TS-RT)<.01 THEN REF=K
2990   IF TS<=50 THEN GOTO 3040
3000   IF TS>50 AND TT=1 THEN GOTO 3040 ELSE GOTO 3020
3010 ' compressor OFF
3020 GDN=7 : WRT$="OF1.2" : GOSUB 5000

```

```

3030 TT=1
3040 '
3050 IF TS>TU THEN GOTO 3070 ELSE 2320
3060 '
3070 GDN=7 : WRT$="OF3" : GOSUB 5000
3075 LOCATE 13,1
3080 PRINT
3090 PRINT "measurement completed.": PRINT
3100 PRINT "data transfer to FDD" : PRINT
3110 '
3120 FOR J=1 TO NF
3130     NZ$=N$+"F"+RIGHT$(STR$(J),1)
3140     OPEN "A:"+NZ$+".DAT" FOR APPEND AS #1
3150     PRINT#1,REF
3160     CLOSE 1
3170 NEXT J
3180 '
3190 PRINT "data transfer completed.":PRINT ← ここで測定とデータ保存終了です。
3195 PRINT"press <PF3> key for menu."
3200 END
3500 '
3600 '
3700 '     ===== subroutine =====
3800 '
5000 ' GBWRT (entry GDN,WRT$) ← GP-IB 制御のためのコードです。
5010 GL=LEN(WRT$)
5020 POKE GCPRM+1,GL+1
5030 POKE GCPRM+2,1
5040 POKE GCPRM+3,GDN
5050 FOR II=1 TO GL
5060     POKE GBUFF+II-1,ASC(MID$(WRT$,II,1))
5070 NEXT II
5080 POKE GBUFF+GL,13
5090 POKE GBUFF+GL+1,10
5100 CALL GPIB(GBWRT)
5110 GOSUB 6800
5120 RETURN
5130 '
6000 ' GBRED (entry GDN,GL return RED$)
6010 POKE GCPRM+1,GL+1
6020 POKE GCPRM+2,1
6030 POKE GCPRM+3,GDN
6040 RED$=""
6050 CALL GPIB(GBRED)
6060 FOR II=1 TO GL
6070     RED$=RED$+CHR$(PEEK(GBUFF+II-1))
6075 '     BF(II)=PEEK(GBUFF+II-1)
6080 NEXT II
6090 GOSUB 6800
6100 RETURN
6110 '

```

・
(途中省略しています。ここは、GP-IB の制御用のサブルーチンになります。)
・

```

6850 '
7000 ' serial polling
7010 CALL GPIB(GBSRD)
7020 POKE GCPRM+1,GDN
7025 CALL GPIB(GBSSTS)
7030 LOCATE 1,1 : PRINT GDN;PEEK (GCPRM) ;PEEK (GCPRM+2)
7040 CALL GPIB(GBSRE)
7050 RETURN
7060 '
7500 ' ← 時間待ちの設定です。
7510 'IF TS=TL THEN L$="30" ELSE L$="15" :' waiting routine
7511 IF TS=TL THEN L$="05" ELSE L$="02" :' waiting routine
7520 T$=TIME$ : HH=VAL(LEFT$(T$,2)) : MM=VAL(MID$(T$,4,2)) :
SS=VAL(RIGHT$(T$,2))
7530 MM=MM+VAL(L$) : HH=HH+MM\60 : MM=MM MOD 60 : DD=DD+HH\24 : HH=HH MOD
24 : HH$=RIGHT$(STR$(HH),2) : MM$=RIGHT$(STR$(MM),2)
7540 IF INSTR(HH$," ")<>0 THEN HH$="0"+RIGHT$(HH$,1)
7550 IF INSTR(MM$," ")<>0 THEN MM$="0"+RIGHT$(MM$,1)
7560 T1$=HH$+" : "+MM$+" : "+RIGHT$(T$,2)
7570 '
7580 '
7590 GDN=7 : WRT$="!" : GOSUB 5000 ← 計測器を制御するための信号を送っています。
7600 GDN=7 : GOSUB 6500
7610 GDN=7 : GL=6 : GOSUB 6000
7620 TR$=RED$
7630 TR=VAL(LEFT$(TR$,6))
7690 GDN=7 : WRT$="MA" : GOSUB 5000
7700 GDN=7 : GOSUB 6500
7710 GDN=7 : GL=6 : GOSUB 6000
7720 TP$=RED$
7730 TP=VAL(LEFT$(TP$,6))
7740 LOCATE 6,21 : PRINT USING"####.# " ;TR
7750 LOCATE 8,21 : PRINT USING"####.# " ;TP
7760 '
7900 T2$=TIME$
7910 IF T2$=T1$ THEN GOTO 7930
7920 GOTO 7570
7930 RETURN
8000 LOCATE 1,10 : PRINT
8010 LOCATE 2,10 : PRINT" Temperature ('C) "
8020 LOCATE 3,10 : PRINT
8030 LOCATE 4,10 : PRINT" Set "
8040 LOCATE 5,10 : PRINT
8050 LOCATE 6,10 : PRINT" Read"
8060 LOCATE 7,10 : PRINT
8070 LOCATE 8,10 : PRINT" Actual"
8080 LOCATE 9,10 : PRINT
8090 LOCATE 10,10 : PRINT" Unit ID"
8100 LOCATE 11,10 : PRINT
8105 RETURN
8110 LOCATE 12,10 : PRINT" Frequency (kHz) "
8120 LOCATE 13,10 : PRINT
8130 LOCATE 14,10 : PRINT
8140 LOCATE 15,10 : PRINT

```

```

8150 LOCATE 16,10 : PRINT" Data Status"
8160 LOCATE 17,10 : PRINT
8170 LOCATE 18,10 : PRINT" C (pF) "
8180 LOCATE 19,10 : PRINT
8190 LOCATE 20,10 : PRINT" D.F.(%) "
8200 LOCATE 21,10 : PRINT
8210 RETURN
8220 '
8510 LOCATE 12,10 : PRINT" "
8520 LOCATE 13,10 : PRINT" "
8530 LOCATE 14,10 : PRINT" "
8540 LOCATE 15,10 : PRINT" "
8550 LOCATE 16,10 : PRINT" "
8560 LOCATE 17,10 : PRINT" "
8570 LOCATE 18,10 : PRINT" "
8580 LOCATE 19,10 : PRINT" "
8590 LOCATE 20,10 : PRINT" "
8600 LOCATE 21,10 : PRINT" "
8610 RETURN
8620 '
9000 ' error
9010 CLOSE
9020 PRINT "Error ";ERR;" in line ";ERL
9030 ON ERROR GOTO 0
9040 STOP
9050 '
9900 '
9910 ' Programmed by Osamu Furukawa
9920 ' Copyright (c) 1989 by Osamu Furukawa
9930 ' All Rights Reserved.

```

なお、同様な形式で、いくつかの変更したバージョンがあります。
これらは、BASIC プログラムのソースコードから装置などの変更点は容易に読み取れると思われま

また、

- ・メニュープログラム
- ・データ表印刷プログラム
- ・データ作図プリント(ハードコピー)プログラム
- ・データ作図プロットプログラム
- ・データ修正プログラム
- ・データ SYLK(Symbolic Link file)変換プログラム
- ・データ転送するプログラム

は、リストを省略させていただきます。

この応用展開例としての、

- ・コンデンサ・誘電体材料の静電容量・誘電損失の温度特性を DC バイアス下で測定するプログラム
- ・コンデンサ・誘電体材料の静電容量・誘電損失を室温で連続測定するプログラム
- ・コンデンサ・誘電体材料の耐電圧の測定プログラム
- ・コンデンサ・誘電体材料の周波数特性の測定プログラム

についても、ここでは割愛させていただきます。

これらの BASIC プログラムのソースコードを希望される場合は、所属、氏名、連絡先、使用目的を、当サイト管理者までお知らせ下されれば幸いです。当方で確認とれしだい、対応させていただきます。